One Click to Create a List of All MEPS-HC ZIP Files' Download URLs: Web Scraping with SAS® and Python

Pradip K. Muhuri¹, Agency for Healthcare Research and Quality Charles Z.X. Han¹, Ryman Healthcare Ltd. John Vickery¹, North Carolina State University

ABSTRACT

This paper presents a case study of web scrapping the Medical Expenditure Panel Survey-Household Component (MEPS-HC) sites with SAS® and Python. Each program dynamically constructs an up-to-date list of over one thousand clickable MEPS-HC ZIP files' download links (Uniform Resource Locators or URLs) and several identifiers and then saves the list into an Excel file. Once you automate the creation of a list of all those URLs in an Excel file using one of these programs (SAS or Python), you can directly initiate ZIP file downloads by simply clicking the relevant URLs from that file outside of the Internet, an easy alternative to navigating multiple pertinent websites for the same task. These programs are reusable to obtain an updated list of MEPS-HC file download links as more data sets are added to the website annually from the ongoing panel survey.

INTRODUCTION

The Agency for Healthcare Research and Quality (AHRQ) has made various data files available from the ongoing Medical Expenditure Panel Survey-Household Component (MEPS-HC²) to the public on its website. On this site (https://meps.ahrq.gov/data_stats/download_data_files.jsp), you can manually select an individual data year or all available years from a dropdown list and check the box that represents the file type(s)³ for your search and then navigate through the site for the desired public-use file (PUF). The PUF site includes links for the data file documentation, survey questionnaires, codebooks, SAS®/Stata/R programs to read the raw data into the individual file format, and ZIP files' download URLs.

Over one thousand ZIP files' download URLs for ASCII, Excel, SAS transport, SAS V9, and Stata files exist across all MEPS-HC PUF websites. However, a consolidated list of all those URLs is unavailable at a single location. Therefore, finding an HTML link to trigger a ZIP file download can be tedious and time-consuming when desiring many such files because the process requires manual navigation across multiple linked websites.⁴ The inefficiency increases with the number of file downloads. The goal of web

¹ Pradip K. Muhuri conceived the idea of the paper, drafted it, and wrote the earlier version of the SAS and Python programs in it. Charles Z.X. Han subsequently revised the Python program and contributed to the code description. Later, John Vickery optimized the SAS and Python programs to their current form. Finally, Pradip K. Muhuri worked on another round of manuscript revisions, to which Charles Z.X and John Vickery contributed.

² MEPS is a set of large-scale on-going surveys of families and individuals, their medical providers, and employers across the United States. It is the complete data source on the cost and use of health care and health insurance coverage for the U.S. civilian noninstitutionalized population [1]. See https://www.meps.ahrq.gov/mepsweb/about_meps/survey_back.jsp for a complete description of MEPS, including its Household Component (HC).

³ Examples of file types include Household Full Year File, Household Event File (e.g., office-based medical provider visits file, outpatient visits file, emergency room visits file, hospital in-patient stays file, home health visits file, dental visits file, and prescribed medicines file), and Pooled Linkage File.

⁴ For the data year 2023, the schedules for public-use file releases spread over ten months from March through December (https://meps.ahrq.gov/mepsweb/about_meps/releaseschedule.jsp).

scraping the MEPS-HC sites with SAS® and Python here is to dynamically construct an up-to-date list of zip files' download URLs and several identifiers and save the list in individual Excel files for easy use.

SAS® PROGRAM

We present the SAS program⁵ (shown in Appendix I) in six steps below.

Step 1: Fetch the content of the main MEPS-HC web page

Lines 3-7: The code performs the HTTP requests to the specified URL. It writes the response body into the fileref SOURCE⁶.

Step 2: Extract data year values by parsing HTML content

Line 11: The DATA statement creates an output SAS data set YEAR VALUES.

Line 14 The INFILE statement specifies the fileref SOURCE and other relevant options (e.g., DLM='<', and EXPANTABS=).

The following are sample data records of the PROC HTTP response.

Line 15: The trailing @ before '<option value" 'and another trailing @ before '>' on the INPUT statement instruct SAS to hold the specific input record and advance the pointer to read the raw data and create a SAS variable YEAR as a numeric variable as specified in the LENGTH statement.

The ?? modifier after YEAR suppresses the invalid data message and prevents the automatic variable _ERROR_ from being set to 1, which results from reading invalid data (e.g., the last sample record above) for the numeric variable. In that case, the variable YEAR will get a missing value.

Line 16: This statement will only output observations with non-missing values for the variable YEAR to the data set YEAR_VALUES.

⁵ We used SAS® 9.04.01M6 X64_10PRO in the Windowing environment.

⁶ Immediately above PROC HTTP, the FILENAME statement associates the file reference (FILEREF) named SOURCE with the external file (directory and filename).

```
NOTE: 3244 records were read from the infile (system-specific pathname).

The minimum record length was 0.

The maximum record length was 452.

NOTE: The data set WORK.YEAR_VALUES has 26 observations and 1 variables.
```

Step 3: Call PROC HTTP and extract public-use file numbers and year URLs via macro processing

Lines 35-77: The macro GET_YEAR with the parameter YEAR wraps PROC HTTP and DATA step in the macro definition. The CALL EXECUTE routine in the DATA _NULL_ step generates the macro calls. With the execution of this macro, the macro variables get resolved. As examples, the resolved values of two macro variables from the first iteration are below.

- &YEAR resolves to 1996
- &YEAR_URL resolves to <u>https://meps.ahrq.gov/data_stats/download_data_files_results.jsp?cboDataYear=1996&buttonYearandDataType=Search</u>

Thus the macro GET_YEAR generates text for 26 PROC HTTP and 26 DATA steps.

We use the PRXPARSE function within the macro definition to create a numbered capture buffer in the string in the DATA step. Consider the Perl regular expressions (PRX) below.

```
prx_puf = prxparse('/<a
href="download_data_files_detail\.jsp\?cboPufNumber=.+\">(.+)<\/a>/');
```

```
prx_meps = prxparse('/<div align="left"
class="contentStyle">(.+)<\/font>/');
```

```
prx_data_year = prxparse('/<div align="left"
class="contentStyle">(.+)<\/font><\/div><\/td>/');
```

In the PRX above, the following characters are escaped with a backslash (\) to perform a match.

- . (literal period, which is by itself a regex metacharacter)
- ? (literal question mark, which is by itself a regex metacharacter)
- / (literal forward slash, which is by itself a metacharacter within regex)

In some of the same PRX, the metacharacter

- .+ matches any one character one or more times
- (.+) matches any one character one or more times from the capture buffer

Lines 46-67: When PRXMATCH finds a match, the call to PRXPOSN uses the results from the PRXPARSE to find the start position (START) and length (END) of the numbered capture buffer within the string. These values are fed into the SUBSTR function to extract values for the PUF number (PUF_NUM), the MEPS file name (MEPS_FILE, code example below), and the data year (DATA_YEAR).

```
meps_file = substr(html_line, start, end);
```

Example output: 2020 Full Year Population Characteristics File

(HC-219 replaced

by HC-224)

Line 59: As you can see, search results for one MEPS file name include strings like
tags in the value of the variable MEPS_FILE (output below). To clean up the stary HTML tag from the results, we use the PRXCHANGE function⁷ below in which the s before the first delimiter(/) indicates that the first regex should be substituted for the second regex. The metacharacters, .+, will match any character that occurs one or more times to find the text to be replaced (i.e.,
br/> in our case). The metacharacter (//) will match null (no character). Thus any character found one or more times in the first regex will be removed and replaced with nothing (i.e., the null regex). This function replaces
br/> (if it exists) with null in the value of the variable MEPS_FILE. The -1 indicates that the function will search and substitute until the end of the input string (MEPS_FILE) is reached [2].

```
meps_file = prxchange('s/<.+>//', -1, meps_file);
```

Example output: 2020 Full Year Population Characteristics File(HC-219 replaced by HC-224)

Finally, the OUTPUT statement writes the observation to the data set YEAR_LIST&YEAR.

81-84: The SAS data sets' names end in numeric year suffixes (e.g., YEAR_LIST_1996, YEAR_LIST_1997. YEAR_LIST_2021). The DATA step concatenates the SAS data sets using the name prefix (YEAR_LIST) followed by a colon(:) to select those ending in numeric suffixes (i.e., YEAR values) in the SET statement. It then filters the data sets whose PUF_NUM begins with 'HC'⁸. The resultant output data set is WORK.YEAR_LIST. Of the various SAS data sets created, the PROC PRINT output from WORK.YEAR_LIST_2021 is below.

PUF_num	meps_file	data_ year
HC-228	2021 Full Year Population Characteristics File	2021
HC-227	2021 Jobs File	2021

⁷ See more about the PRXCHANGE function and its arguments in the SAS® Documentation: https://documentation.sas.com/doc/en/vdmmlcdc/8.1/lefunctionsref/n0r8h2fa8djqf1n1cnenrvm573br.htm.

⁸ PUF_NUMs not beginning with 'HC' include 26 NHIS Link Files, 15 NHEA-Aligned MEPS Files, and eleven NHC_: and LINK_: files (the colon is a wild character) as of this writing.

Lines 87-89: The NODUPKEY option in the PROC SORT statement creates an output data set YEAR_VALUES with no duplicate observations (i.e., only one observation for the _ALL_ variable values).

The partial SAS Log9:

```
NOTE: There were 509 observations read from the data set WORK.YEAR_LIST.

NOTE: 79 observations with duplicate key values were deleted.

NOTE: The data set WORK.YEAR_LIST has 430 observations and 3 variables.

NOTE: PROCEDURE SORT used (Total process time):
```

Step 4: Call PROC HTTP and extract public-use file (PUF) URLs to construct ZIP file URLs via macro processing

Lines 92-139: The macro GET_PUF with one positional parameter (PUFNUM) wraps PROC HTTP and DATA step in the macro definition. The CALL EXECUTE routine in the DATA_NULL_ step generates the macro calls. With the execution of this macro, the macro variables get resolved. As examples, the resolved values of three macro variables from the first iteration are below.

- &PUFNUM resolves to HC-092
- &PUF_URL resolves to <u>https://meps.ahrq.gov/mepsweb/data_stats/download_data_files_detail.jsp?cboPufNumber=HC0</u>
 92
- PUF_LIST_%sysfunc(translate(&PUFNUM, '_', '-')) resolves to PUF_LIST_HC_092

This macro GET_PUF generates text for hundreds of URL schemes to create HTTP requests and over one thousand DATA steps to construct ZIP files' download URLS.

We use the PRXPARSE function within the macro definition to create a numbered capture buffer in the string in the DATA step. Consider the Perl regular expressions (PRX) below.

```
prx_data_file = prxparse('/Data File.*, (.+)<\/td>/');
```

```
prx_zip = prxparse('/<a href="\.\.\/(data_files[\/pufs]*\/.+\.zip)">ZIP<\/a>/');
```

⁹ The duplicate values include two or more occuurences of the MEPS Longitudinal Data File from various panels, the 2000-2013 Employment Variables File, the 2002-2009 Risk Adjustment Scores File, the 1996-2001 Risk Adjustment Scores File (HC-081 replaced by HC-092), the 2001 and 2002 MEPS HC Survey Data (CD-ROM), the Multum Lexicon Addendum Files to MEPS Prescribed Medicines Files 1996-2013, and the 1999 and 2000 MEPS HC Survey Data (CD-ROM).

When PRXMATCH finds a match, the call to PRXPOSN uses the results from the PRXPARSE to find the start position (START) and length (END) of the numbered capture buffer within the string. These values are fed into the SUBSTR function to extract values for the PUF number (PUF_NUM) and the file format (FILE FORMAT). Note using the CATX and SUBSTR functions to create the ZIP file link (ZIP LINK).

Lines 142-144: The DATA step concatenates the SAS data sets¹⁰ using the name prefix (PUF_LIST) followed by a colon(:) to select those ending PUF numbers as suffixes. The resultant output data set is WORK.PUF_LIST.

The next is the PROC PRINT output from WORK.ZIP_LIST_HC_228.

PUF_num	file_format	zip_link
HC-228	ASCII format	https://meps.ahrq.gov/data_files/pufs/h228/h228dat.zip
HC-228	SAS transport format	https://meps.ahrq.gov/data_files/pufs/h228/h228ssp.zip
HC-228	SAS V9 format	https://meps.ahrq.gov/data_files/pufs/h228/h228v9.zip
HC-228	Stata format	https://meps.ahrq.gov/data_files/pufs/h228/h228dta.zip
HC-228	XLSX format	https://meps.ahrq.gov/data_files/pufs/h228/h228xlsx.zip

Lines 147-149: The NODUPKEY option in the PROC SORT statement creates an output data set PUF_LIST with no duplicate observations (i.e., only one observation for the _ALL_ variable values).

The partial SAS Log:

```
NOTE: There were 1179 observations read from the data set WORK.PUF_LIST.

NOTE: 0 observations with duplicate key values were deleted.

NOTE: The data set WORK.PUF_LIST has 1179 observations and 3 variables.
```

Step 5: One-to-many matching - join up extracted pieces next to each other

We combine extracted data elements horizontally. It is like joining knitted pieces.

Lines 152-159: The PROC SQL performs a one-to-many match with the tables YEAR_LIST and PUF_LIST based on the key column PUF_NUM and creates a table MEPS_ZIP_LINKS. Note the following.

- The table YEAR LIST has a unique value for the column PUF NUM.
- The table PUF_LIST includes multiple rows with the same value as the column PUF_NUM.
- The output table MEPS_ZIP_LINKS has a unique value for the column ZIP_LINK.

¹⁰ A few SAS data sets with missing values on the variable ZIP_LINK are empty due to the replacement of the ZIP file with an updated version along with a different PUF number and URL on the webaite; however, the old PUF URL still exists on the Internet with no embedded hyperlinks pointing to file download URLs. For example, see

https://meps.ahrq.gov/mepsweb/data_stats/download_data_files_detail.jsp?cboPufNumber=HC-002 (HC-012 replaced HC-002).

NOTE: Table WORK.MEPS ZIP LINKS created, with 1179 rows and 5 columns.

Step 6: Create listings with PROC REPORT and output them into an Excel spreadsheet

Lines 167-183: The REPORT procedure identifies the final merged data set, displaying the values of all specified variables. The CALL DEFINE associates the URL with ZIP_LINK. The ODS EXCEL statement directs the PROC REPORT output to a file listed in the FILE= option. With the execution of the ODS EXCEL CLOSE; statement, the Excel file gets created (part of the output from the Excel spreadsheet shown in APPENDIX II).

The combined partial SAS Log:

NOTE: There were 1179 observations read from the data set WORK.MEPS_ZIP_LINKS NOTE: Writing EXCEL file: c:\SESUG 2023\SAS MEPS zip links 2023-06-07.xlsx

PYTHON PROGRAM

We used Python (Version 3.9.12) in the Spyder IDE and JupyterLab Python Notebook (Version 6.4.5) as our second solution for web scraping. We present the Python program (as shown in APPENDIX III) in four steps below.

Step 1: Import libraries

Lines 8-11: The program imports necessary libraries, including Requests and BeautifulSoup¹¹ to scrape and parse MEPS-HC websites and dynamically create a list of over one thousand URLs that would trigger data downloads.

Step 2: Get the list of numerical year options and saving as the list of tuples with year and URL

Lines 18-19: This code uses the 'requests' library to send a GET request to a website with the URL "https://meps.ahrq.gov/data_stats/download_data_files.jsp". The response from the website is then passed to BeautifulSoup, another library, which parses the HTML text into an object called main_soup.

Line 23: The code defines a base URL as

"https://meps.ahrq.gov/mepsweb/data_stats/download_data_files_results.jsp?cboDataYear=", used to construct the URLs for downloading data files.

Line 24-26: The 'year_url_suffix' variable is set to "&buttonYearandDataType=Search", which will be appended to each year-specific URL. Next, the code extracts the available years for data download from 'main_soup'. It does so by finding the first 'select' element with the name attribute set to "cboDataYear", which is used to choose a specific year for data download. The '.select("option")' method is then called on

¹¹ BeautifulSoup, Scrapy, and Selenium are the three main Python-based tools commonly used for web scraping. BeautifulSoup() of the Python bs4 library enables one to extract data (specific elements) from a single webpage at a time effectively. Scrapy is a web scraping framework that can crawl various web pages, downloading, parsing, and storing data, whereas Selenium can automate navigating to sites and scrape the webpage content dynamically.

this `select` element, which returns all the 'option' elements. The 'year_options' variable now contains a list of all the available years for data download.

The code below prints the list year options.

```
print(year_options)
```

The output:

[<option value="All">All available years</option>, <option value="2021">2021</option>, <option value="2020">2020</option>, <option value="2019">2019</option>, <option value="2018">2018</option>, <option value="2017">2017</option>, <option value="2017">2017</option>, <option value="2016">2016</option>, <option value="2015">2015</option>, <option value="2014">2014</option>, <option value="2010">2010</option>, <option value="2010">2010</option>, <option value="2009">2009</option>, <option value="2008">2008</option>, <option value="2007">2007</option>, <option value="2006">2006</option>, <option value="2005">2005</option>, <option value="2004">2004</option>, <option value="2003">2003</option>, <option value="2002">2002</option>, <option value="2001">2001</option>, <option value="2001">2001</option>, <option value="2001">2001</option>, <option value="1999">1999</option>, <option value="1998">1998</option>, <option value="1997">1997</option>, <option value="1996">1996</option>, <option value="Projected Years">Projected Years</option>]

Lines 28-31 The code creates a 'year_url_list' by iterating through each "option" element in 'year_options'. If the "value" attribute of the "option" element is a digit, a tuple is added to 'year_url_list'. The first element of the tuple is the value of the "value" attribute, representing the year for data download. The second element is a URL constructed by concatenating the base URL with the year value, passed through the 'quote()' method to ensure it is appropriately encoded for use in a URL. Finally, the 'year url suffix' variable is appended to the URL.

Overall, this code retrieves available data years associated with the public-use file name from the website, constructs URLs for downloading data files for each year, and stores them in a list of tuples called 'year_url_list'.

Below is the additional code (not shown in the Python program in Appendix III) to print the first two elements in the list year_url_list.

```
for item in year_url_list[:2]: print(item)
```

The output:

```
('2021', 'https://meps.ahrq.gov/mepsweb/data_stats/download_data_files_results.j
sp?cboDataYear=2021&buttonYearandDataType=Search')

('2020', 'https://meps.ahrq.gov/mepsweb/data_stats/download_data_files_results.j
sp?cboDataYear=2020&buttonYearandDataType=Search')
```

Step 3: Get the list (i.e., year_url_list) of tuples to obtain "HC-" zip file links

Lines 38-39: The `base_hc_url` variable remains the same, representing the base URL for constructing Zip files' download URLs. The `zip_link_list` is an empty list that will store dictionaries containing information about the ZIP file download links.

Lines 40-43: The code then enters a `for` loop that iterates through each tuple in `year_url_list`. Inside the loop, it sends a GET request to the year-specific URL, and if the request is successful (`response.raise for status()`), it proceeds to extract the data using `pd.read html()`.

Lines 45-46: The `pd.read_html()` function parses the HTML response text and attempts to extract tables from it. Each table is represented as a DataFrame in the `dfs` list. Next, the code initializes an empty DataFrame called `hc_df` to store the relevant data from the tables.

Lines 47-53: For each DataFrame `df` in `dfs`, the code checks if it contains the column "File(s), Documentation & Codebooks". If it does, the code concatenates `df` with `hc_df`, drops rows with missing values in the "PUF no." column, selects only rows where the "PUF no." contains "HC-", drops duplicate rows based on the "PUF no.", and appends the MEPS PUF URL by combining `base_hc_url` with the values in the "PUF no." column.

Lines 54-66: If `hc_df` is not empty, the code renames the columns of `hc_df` and iterates over each row using `hc_df.itertuples()`. For each row, it sends a GET request to the healthcare data file URL (`row.hc_url'), raises an exception if the request fails, and parses the HTML response using `BeautifulSoup`.

Lines 68-83: If successful, the code extracts the name of the MEPS-HC PUF file from the HTML using `hc_soup.find(class_="OrangeBox").text` and searches for a specific table cell ("td") containing the text "Data File". If found, it creates a dictionary called `zip_link_dict` and populates it with information such as the PUF number, data year, PUF name, file format, and the ZIP file download link. The `zip_link_dict` is then appended to `zip_link_list`.

The code below prints selected elements from `zip_link_list`.

```
for item in zip_link_list[:2]: print(item)
```

The output:

```
{'PUF_num': 'HC-228', 'data_year': '2021', 'meps_file': 'MEPS HC-228: 2021 Full Year Popul ation Characteristics File', 'file_format': 'Data File, ASCII format', 'zip_link': 'https://meps.ahrq.gov/data_files/pufs/h228/h228dat.zip'}

{'PUF_num': 'HC-228', 'data_year': '2021', 'meps_file': 'MEPS HC-228: 2021 Full Year Popul ation Characteristics File', 'file_format': 'Data File, SAS transport format', 'zip_link': 'https://meps.ahrq.gov/data_files/pufs/h228/h228ssp.zip'}
```

Lines 84-90: Any exceptions that occur during the execution of the code are caught using `try`/ except` blocks and error messages are printed.

Overall, this code retrieves ZIP files' download URLs for each available year based on the information obtained from the previous code and appends the relevant information to the `zip_link_list`.

Step 4: Create an Excel file from the Pandas DataFrame

Line 100: The code first creates the DataFrame `meps_df` using the `pd.DataFrame()` constructor and passing `zip_link_list` as an argument, which contains the information about the ZIP file download links.

Line 101: The code removes duplicate rows in `meps_df` using the `drop_duplicates()` method with the `inplace=True` argument to modify the DataFrame in-place.

Line 102: The code modifies the "meps_file" column in `meps_df` by splitting the values using `str.split(", n=1)` and selecting the second part of the split using `str[-1]`. It extracts the MEPS file name and updates the values in the "meps_file" column.

Line 103: The code modifies the "file_format" column in `meps_df` by splitting the values using `str.split(", ", n=1)` and selecting the second part of the split using `str[-1]`. It extracts the file format and updates the values in the "file_format" column.

Line 104: The code performs multi-key sorting using the 'sort_values' attribute with the ascending= argument using True or False for the same number of values.

Line 105: The code prints the first few rows of the modified DataFrame using the `head()` method.

```
meps_file
                                           file format \
data_year puf_num
20
        2021 HC-227 2021 Jobs File
                                             ASCII format
22
       2021 HC-227 2021 Jobs File
                                            SAS V9 format
21
       2021 HC-227 2021 Jobs File SAS transport format
23
       2021 HC-227 2021 Jobs File
                                             Stata format
24
        2021 HC-227 2021 Jobs File
                                            XLSX format
                                            zip_link
20 https://meps.ahrq.gov/data files/pufs/h227/h22...
22 https://meps.ahrq.gov/data_files/pufs/h227/h22...
21 https://meps.ahrq.gov/data_files/pufs/h227/h22...
23 https://meps.ahrq.gov/data_files/pufs/h227/h22...
24 https://meps.ahrq.gov/data files/pufs/h227/h22...
```

Line 109: The variable `today` represents the current date. It uses `pd.Timestamp("now")` to obtain the current timestamp and `strftime("%Y-%m-%d")` to format it as "YYYY-MM-DD".

Line 110: The `to_excel()` method is used to write the DataFrame to an Excel file, with the `index=False` argument to exclude the DataFrame index from the output. The code saves the DataFrame to an Excel file with a dynamically generated filename. It uses f-string formatting to include the `today` variable in the filename, resulting in a filename like "C:\SESUG_2013\Python_Solution_WS_YYYY-MM-DD.xlsx".

Overall, this code modifies and displays the DataFrame `meps_df`, generates the current date, and saves the DataFrame to an Excel file with a filename that includes the current date in the "output" folder.

Comparing SAS and Python Output

We read both SAS and Python-generated Excel files using read_excel() method of pandas into pandas DataFrames. We then compare the two DataFrames using pandas compare(). This comparison confirms that SAS and Python applications' file download URL lists are identical.

```
#!/usr/bin/env python
# coding: utf-8

import pandas as pd
df_sas = pd.read_excel(f"C:\SESUG_2023\SAS_Solution_WS_2023-06-10.xlsx")
df_py = pd.read_excel(f"C:\SESUG_2023\Python_Solution_WS_2023-06-10.xlsx")
diff = df_sas.compare(df_py, keep_equal=True, keep_shape = True)
print(diff)
```

The partial output:

```
zip link
     data_year
                                                                 other
          self ...
0
          2021 ... https://meps.ahrq.gov/data_files/pufs/h227/h22...
          2021 ... https://meps.ahrq.gov/data_files/pufs/h227/h22...
1
2
          2021 ...
                    https://meps.ahrq.gov/data files/pufs/h227/h22...
3
          2021
                    https://meps.ahrq.gov/data files/pufs/h227/h22...
               . . .
                ... https://meps.ahrq.gov/data_files/pufs/h227/h22...
4
          2021
         1996 ...
1174
                     https://meps.ahrq.gov/data files/pufs/h17ssp.zip
1175
         1996
                     https://meps.ahrq.gov/data files/pufs/h24dat.zip
               . . .
         1996 ...
                    https://meps.ahrq.gov/data_files/pufs/h24ssp.zip
1176
1177
         1996 ... https://meps.ahrq.gov/data files/pufs/h41dat.zip
1178
         1996 ...
                     https://meps.ahrq.gov/data files/pufs/h41ssp.zip
[1179 rows x 10 columns]
```

CONCLUSION

The paper offers SAS® and Python solutions to dynamically create a list of MEPS-HC files' download URLs representing various formats (e.g., ASCII, Excel, SAS transport, SAS V9, and Stata files). This comparison confirms that SAS and Python applications' file download URL lists are identical This work found that code efficiencies varied between the two programs. For example, Python's libraries BeautifulSoup and urllib.request for HTTP Requests came in handy to perform web scraping. On the other hand, in performing the same task, some users may find the SAS program much more intuitive. However, we see the Python solution as the most efficient due to its flexibility, ease of use, and minimal coding.

These programs are reusable to obtain an updated list of MEPS-HC ZIP files' download links as more data sets are added to the website annually from the ongoing panel survey. In other words, future use of our programs will likely result in more such URLs depending on the timing of their execution, as the target web page and the linked websites get updated with new files. Once you automate the creation of a list of all MEPS-HC ZIP files' download URLs in an Excel file using one of these programs (SAS or Python), you

can directly initiate ZIP file downloads by simply clicking the relevant URLs from that file outside of the Internet, an easy alternative to navigating multiple pertinent websites for the same task.

REFERENCES

- Cohen, J., S. Cohen, and J. Banthin. 2009. "The Medical Expenditure Panel Survey: A National Information Resource to Support Health Care Cost Research and Inform Policy and Practice." Medical Care, 47:S44-50.
- 2. Alabaster, A, and M.A. Armstrong. 2020. "Cracking Cryptic Doctors' Notes with SAS® PRX Functions. SAS Global Forum Paper 2020.

ACKNOWLEDGMENTS

The views expressed in this paper are those of the authors, and no official endorsement by the Health and Human Services or the Agency for Healthcare Research and Quality (AHRQ) is intended or should be inferred. However, the authors thank AHRQ staff and Kurt Bremser in SAS Support Communities for reviewing earlier manuscript versions and providing valuable comments. Finally, the authors are responsible for any errors or omissions.

CONTACT INFORMATION

Your comments are valued and encouraged. Contact the authors at:

Pradip K. Muhuri, PhD AHRQ/CFACT 5600 Fishers Lane Rockville, MD 20857 Pradip.Muhuri@ahrq.hhs.gov

Charles Z.X. Han Ryman Healthcare Ltd. charleszhouxiaohan@gmail.com

John Vickery North Carolina State University Libraries (919) 513-0344 John_vickery@ncsu.edu

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

```
/* APPENDIX I (SAS Program) */
1 %let path = c:\SESUG 2023;
2 /* Step 1: Fetch the main MEPS-HC web page's contents using PROC HTTP */
3 filename source temp;
4 proc http
     url = "https://meps.ahrq.gov/data stats/download data files.jsp"
5
    out=source;
6
7 run;
8
9 /* Step 2: Parse the MEPS-HC main web page */
10 /* Get years from dropdown options list */
11 options generic;
12 data year values;
13
     length year 8;
14
     infile source dlm='<' expandtabs;
     input @'<option value="' @'>' year ??;
15
16 if not missing (year);
17 run;
18
19 filename source clear;
20
21 /* Step 3: Call Proc HTTP for each year url */
22 %macro get year(year);
23
       filename yearresp "&path\hc response.txt";
       %local base url data year param search param year url;
24
25
       /* URL elements */
       %let base_url = https://meps.ahrq.gov/data_stats/download_data_files_results.jsp?;
26
27
       %let data_year_param = cboDataYear=&year.;
28
       %let search param = %nrstr(&buttonYearandDataType=Search);
29
30
       /* Combine the URL elements to construct the year url*/
       %let year url = &base url.&data year param.&search param.;
31
32
       /* Call PROC HTTP to retrieve the content */
33
34
       /* of each year search results */
35
       proc http
               url = "&year_url."
36
37
       out=yearresp;
38
       run;
39
40
       data year_list_&year. (keep=puf_num meps_file data_year);
               length puf num $15 meps file $150 data year $10;
41
42
               infile yearresp length = reclen lrecl = 32767 end=eof;
               /* regex to get the PUF num in the table of results */
43
               prx puf = prxparse('/<a</pre>
44
href="download_data_files_detail\.jsp\?cboPufNumber=.+\">(.+)<\/a>/');
               /* regex to get the meps file */
45
               prx_meps = prxparse('/
46
class="bottomRightgrayBorder"><div align="left" class="contentStyle">(.+)<\/font>/');
```

```
/* regex for data year */
47
               prx_data_year = prxparse('/
48
class="bottomRightgrayBorder"><div align="left" class="contentStyle">(.+)<\/font><\/div><\/td>/');
               /* Read the HTML line by line */
49
50
               do while (not eof);
51
                       input html line $varying32767. reclen;
52
                       if prxmatch(prx_puf, html_line) > 0 then do;
53
                              call prxposn(prx puf, 1, start, end);
54
                       puf_num = substr(html_line, start, end);
                       end;
55
56
                 if prxmatch(prx_meps, html_line) > 0 then do;
57
                       call prxposn(prx meps, 1, start, end);
58
                       meps file = substr(html line, start, end);
59
                       meps_file = prxchange('s/<.+>//', -1, meps_file); /* remove any stray html tags
*/
60
                 end;
                 if prxmatch(prx data year, html line) > 0 then do;
61
62
                              call prxposn(prx_data_year, 1, start, end);
63
                       data year = substr(html line, start, end);
64
                       output year_list_&year.;
65
                       end;
66
               end;
67
       run;
68
69
       filename yearresp clear;
70
71 %mend get year;
72
73 /* Loop through each year of the year values dataset and call the macro */
74 data _null_;
75
       set year values;
76
     call execute('%nrstr(%get_year('||strip(year)||'));');
77 run;
78
79 /* Concatenate all year list YEAR datasets */
80 /* Just keep obs with puf num beginning HC- */
81 data year list;
82
       set year list :;
83
       if substr(puf num, 1, 2) = 'HC';
84 run;
85
86 /* De-dup */
87 proc sort data=year_list nodupkey;
88 by _ALL_;
89 run;
90
91 /* Step 4: Get the zip files for each HC- puf num. */
92 %macro get puf(pufnum);
93
     %local base url puf url;
```

```
94
       filename pufresp temp;
95
96
       /* URL elements */
97
       %let base url =
https://meps.ahrq.gov/mepsweb/data stats/download data files detail.jsp?cboPufNumber=;
98
99
       /* Combine the URL elements to construct the PUF url */
100
       %let puf url = &base url.&pufnum.;
101
102
       proc http
103
               url="&puf_url."
104
               out= pufresp;
105
       run;
106
       data puf_list_%sysfunc(translate(&pufnum, '_', '-')) (keep=puf_num file_format zip_link);
107
108
               length puf num $15 file format $150 zip link $200;
109
               infile pufresp length = reclen lrecl = 32767 end=eof;
110
               prx_data_file = prxparse('/<td width="50%" height="0"
111
class="bottomRightgrayBorder">Data File.*, (.+)<\/td>/');
112
         prx_zip = prxparse('/<a href="\.\.\/(data_files[\/pufs]*\/.+\.zip)">ZIP<\/a>/');
113
114
               do while (not eof);
                       input html line $varying32767. reclen;
115
116
                       puf num = "&pufnum.";
117
                       if prxmatch(prx data file, html line) > 0 then do;
                               call prxposn(prx data file, 1, start, end);
118
                       file format = substr(html line, start, end);
119
120
                       end;
121
122
                       if prxmatch(prx zip, html line) > 0 then do;
123
                               call prxposn(prx_zip, 1, start, end);
124
                       zip_link = cats("https://meps.ahrq.gov/", substr(html_line, start, end));
125
                       output puf_list_%sysfunc(translate(&pufnum, '_', '-'));
126
                       end;
127
               end;
128
129
       run;
130
131
       filename pufresp clear;
132
133 %mend get puf;
134
135 /* Loop through each puf num of the year list dataset and call the macro */
136 data _null_;
137
       set year list;
       call execute('%nrstr(%get_puf('||strip(puf_num)||'));');
138
139 run;
140
```

```
141 /* Concatenate all puf list datasets */
142 data puf_list;
143
       set puf list :;
144 run;
145
146 /* De-dup */
147 proc sort data=puf_list nodupkey;
148
       by _ALL_;
149 run;
150
151 /* Step 5: Merge year_list and puf_list */
152 proc sql;
153
       create table meps zip links as
154
       select y.puf_num, y.meps_file, y.data_year, p.file_format, p.zip_link
155
       from year list as y,
156
                puf list as p
       where y.puf num = p.puf num
157
158
      order by data_year desc, puf_num, file_format;
159 quit;
160
161 /* Format the current date as "YYYY-MM-DD" */
162 %let current_date = %sysfunc(today());
163 %let formatted_date = %sysfunc(putn(&current_date., yymmdd10.));
164
165 /* Step 6: Direct proc report output to excel */
166 ods listing close;
167 ods excel file = "&path\SAS Solution WS &formatted date..xlsx"
168 options (sheet name = 'Sheet1'
169 flow="header,data" row_heights = '15'
     absolute_column_width='11,11,70,30,55');
171 proc report data=meps zip links;
172 column data_year puf_num meps_file file_format_zip_link;
173 define puf num / display;
174 define meps_file / display;
175 define data year / display;
176 define file format / display;
177 define zip link / display;
178 compute zip link;
179
      call define( col ,"url",zip link);
180 endcomp;
181 run;
182 ods excel close;
183 ods listing;
```

APPENDIX II (Part of the Excel file from the PROC REPORT Output)

data_year	PUF_num	meps_file	file_format	zip_link
2021	HC-228	Full Year Population Characteristics File	ASCII format	https://meps.ahrq.gov/data_files/pufs/h228/h228dat.zip
2021	HC-228	Full Year Population Characteristics File	SAS V9 format	https://meps.ahrq.gov/data_files/pufs/h228/h228v9.zip
2021	HC-228	Full Year Population Characteristics File	SAS transport format	https://meps.ahrq.gov/data_files/pufs/h228/h228ssp.zip
2021	HC-228	Full Year Population Characteristics File	Stata format	https://meps.ahrq.gov/data_files/pufs/h228/h228dta.zip
2021	HC-228	Full Year Population Characteristics File	XLSX format	https://meps.ahrq.gov/data_files/pufs/h228/h228xlsx.zip
2021	HC-227	Jobs File	ASCII format	https://meps.ahrq.gov/data_files/pufs/h227/h227dat.zip
2021	HC-227	Jobs File	SAS V9 format	https://meps.ahrq.gov/data_files/pufs/h227/h227v9.zip
2021	HC-227	Jobs File	SAS transport format	https://meps.ahrq.gov/data_files/pufs/h227/h227ssp.zip
2021	HC-227	Jobs File	Stata format	https://meps.ahrq.gov/data_files/pufs/h227/h227dta.zip
2021	HC-227	Jobs File	XLSX format	https://meps.ahrq.gov/data_files/pufs/h227/h227xlsx.zip

```
Appendix III (Python program)
1 #!/usr/bin/env python
2 # coding: utf-8
3
4 # **step 1**
5 # - note the python libraries/modules
7 # import libraries
8 import pandas as pd
9 import requests
10 from bs4 import BeautifulSoup
11 from urllib.parse import quote
12
13 # **step 2**
14 # - get the list of numerical year options from the main page
15 # - save as list of tuples with year and url
16
17 # main page response
18 main_page = requests.get("https://meps.ahrq.gov/data_stats/download_data_files.jsp")
19 main_soup = BeautifulSoup(main_page.text, "html.parser")
21 # get list of data years and links for each year
22 # skip the "All years" (i.e. non-digit options)
23 base year url =
"https://meps.ahrq.gov/mepsweb/data stats/download data files results.jsp?cboDataYear="
24 year url suffix = "&buttonYearandDataType=Search"
26 year_options = main_soup.select_one('select[name="cboDataYear"]').select("option")
27
28 year_url_list = [
      (x.get("value"), base_year_url + quote(x.get("value")) + year_url_suffix)
29
30
     for x in year options
      if x.get("value").isdigit()
31
32 ]
33
34 # **step 3**
35 # - use the year_url_list list of tuples to get "HC-" zip file links
36 # - save these in list of dictionaries
37
38 base hc url =
"https://meps.ahrq.gov/mepsweb/data_stats/download_data_files_detail.jsp?cboPufNumber="
39 zip link list = []
40 for year, year_url in year_url_list:
41
     try:
        response = requests.get(year url)
42
        response.raise_for_status()
43
44
45
           dfs = pd.read html(response.text)
          hc_df = pd.DataFrame()
46
47
          for df in dfs:
48
             if "File(s), Documentation & Codebooks" in df.columns:
49
                hc_df = pd.concat([hc_df, df], ignore_index=True)
```

```
50
                 hc df = hc df.dropna(subset="PUF no.")
                 hc df = hc df.loc[hc df["PUF no."].str.contains("HC-")]
51
52
                 hc df = hc df.drop duplicates(subset="PUF no.")
53
                 hc_df["hc_url"] = base_hc_url + hc_df["PUF no."]
54
           if not hc_df.empty:
55
              hc df.columns = [
56
                 "puf_num",
57
                 "Files",
58
                 "Data Update",
59
                 "Year",
60
                 "File_Type",
61
                 "hc_url",
62
63
              for row in hc df.itertuples():
64
                 hc response = requests.get(row.hc url)
65
                 hc response.raise for status()
66
                 hc_soup = BeautifulSoup(hc_response.text)
67
68
                    meps_file = hc_soup.find(class_="OrangeBox").text
69
                    for td in hc_soup.find_all("td"):
70
                      zip link dict = {}
71
                      if td.text.startswith("Data File"):
72
                         zip_link_dict["data_year"] = row.Year
                         zip_link_dict["puf_num"] = row.puf num
73
74
                         zip link dict["meps file"] = meps file
75
                         zip_link_dict["file_format"] = td.text
76
                         zip link dict[
77
                            "zip link"
                         ] = "https://meps.ahrq.gov" + td.find_next("a").get(
78
79
                           "href"
80
                         ).strip(
81
82
83
                         zip_link_list.append(zip_link_dict)
84
                 except:
85
                    # catch your exceptions here
86
87
         except requests.exceptions.HTTPError as err:
88
            print(err)
      except requests.exceptions.HTTPError as httperr:
89
90
         print(httperr)
91
92 # **step 4**
93 # - save the list of dictionaries to pandas dataframe
94 # - deduplicate
95 # - clean up meps_file column to remove "PUF no." prefix
96 # - clean up file format column to remove "Data File" prefix
97 # - sort the dataframe in certain order
98 # - get n rows of the dataframe
99
100 meps df = pd.DataFrame(zip link list)
101 meps_df.drop_duplicates(inplace=True)
102 meps_df["meps_file"] = meps_df["meps_file"].str.split(": ", n=1).str[-1]
103 meps_df["file_format"] = meps_df["file_format"].str.split(", ", n=1).str[-1]
104 meps_df = meps_df.sort_values(by=['data_year','puf_num', 'file_format'], ascending=[False, True,
True])
105 print(meps_df.head(5))
```

```
106
107 #**step 4 (continued)**
108 # - obtain the current date (timestamp)
109 # - save the dataframe as an excel file
110
111 today = pd.Timestamp("now").strftime("%Y-%m-%d")
112 meps_df.to_excel(f"C:\SESUG_2023\Python_Solution_WS_{today}.xlsx", index=False)
```